Painting

The first part of our cartoon renderer is the painter.  As a minimum, a painter needs to support rendering an object with solid colors.  A more advanced painter will generate stepped shadows and highlights using light direction, vary this shading with brightness, and even support multiple light sources. 

Displaying objects with solid colors is rather trivial, but some research has been done into different stepped shading techniques.  According to Claes [1], it is possible to choose a constant color (either highlight or shadow) for each polygon, but better results occur if the polygons along shadow-boundaries are subdivided using linear interpolation.  In order to use programmable hardware to achieve a similar effect, it is necessary to use the light direction vector and vertex normal to look up a texture coordinate from a specialized shading texture.  This is similar to the algorithm used by Lake [2].

The simplest hardware implementation of a painting algorithm that supports stepped shading uses a one-dimensional texture.  In listing 1, the vertex shader takes the direction of a single light as its input and calculates the dot product between the light and the vertex normal.  The dot product of two normalized vectors ranges from -1 to 1.  A value of one means that the two vectors point in exactly the same direction, but as the angle between the two vectors increases, the value will decrease.  We can convert the result of the dot product into a texture coordinate and clamp the negative values to zero.  The right side (u=1.0) of the texture will represent normals that are facing the light and should be bright, and the left side (u=0.0) will represent normals that are facing away from the light and should be dark.

Listing 1

vs.1.1

dcl_normal v3

; transform normal

dp3 r1.x, v3, c[INVERSE_WORLD_MATRIX]

dp3 r1.y, v3, c[INVERSE_WORLD_MATRIX_1]

dp3 r1.z, v3, c[INVERSE_WORLD_MATRIX_2]

; renormalize it

dp3 r1.w, r1, r1

rsq r1.w, r1.w

mul r1, r1, r1.w

; dot product normal and light into first texture stage

dp3 oT0, r1, c[LIGHT_POSITION]
We can sample a one dimensional texture with the coordinates from the vertex shader and modulate it with the color of the object.  This color can be a constant that is loaded from material properties, or it can be an actual texture.  If we used a gradient as our one-dimensional texture, we would get something very similar to realistic shading, but the beauty of this method comes when we use a specialized one-dimensional texture that gives us a stepped shading effect.

Our one-dimensional texture has an initial area that is a dark gray, followed by a lighter area.  The advantage of putting the shadows and highlights in a texture like this are two-fold.  Firstly, it protects us from having to conditionally choose between different shades in the pixel shader, a slow and difficult process with PS 1.1.  Secondly, it makes it very easy to change the shading style.  To lower the contrast of the highlight and shadow areas, simply use more similar shades of gray.  If three or four highlight shades are desired, one needs only to add more steps in the texture.

Our first implementation looks quite good for smooth, high-polygon models, but breaks down when the models are low-polygon or have large flat surfaces.  Because the texture coordinates are calculated per-vertex, the linear interpolation over the faces of the triangles can give the edges of the shadows an angular look.  As we know, our hardware has the capacity for per-pixel lighting, and we can use a modified form of this to create a per-pixel version of our stepped shading algorithm.

The per-pixel algorithm needs to do the dot-product inside the pixel shader. This way, a different lighting value can be acquired for each pixel.  This result can then be used to do a 1D texture lookup as in the previous algorithm.  While PS 1.2 gives us an excellent tool for this in the texdp3tex instruction, PS1.1 has only a set of dual instructions to do what we need, the texm3x2pad and texm3x2tex instructions.

Listing 2

vs.1.1 ; modifications to previous vertex shader

; stick normal vector in first stage

mov oT0, r1

; stick light position in second stage

mov oT1, c[LIGHT_POSITION]

ps.1.1

def c1, 0.0f, 0.0f, 0.0f, 1.0f
; black

texcoord t0



; grab the normal

texm3x2pad t1, t0_bx2

; dot with light for u value

texm3x2tex t2, t0_bx2

; grab shading texture

tex t3




; regular texture data

mov r0, C_MATERIAL_DIFFUSE
; load constant color in

cnd r0, r0.a, r0, t3

; use color if alpha







; or else use texture

mul r0.rgb, r0, t2


; modulate with shading

Listing 2 shows a pixel shader that implements the per-pixel algorithm.  A vertex shader places the light direction in the second set of texture coordinates. It also biases the vertex normal into the 0 to 1 range and places it in the first set of coordinates.  It is necessary to bias the normal because we will be sampling it with texcoord, which clamps the resulting values to 0 to 1, but, fortunately, we have an easy way to unbias the values with the _bx2 instruction modifier. After we have retrieved the normal with texcoord, we use the texm3x2 pair to grab the dot-product with the light normal and lookup the one-dimensional texture from the third stage.

As may be obvious from the name, the texm3x2 instructions actually do two dot-product calculations.  The first dot product is done with the second stage coordinates and gives us a u-coordinate for our sampled texture.  The second dot product uses the third stage to create a v-coordinate, which is completely unused in our one-dimensional lookup.  It seems silly to waste a whole calculation on every pixel, and, in fact, we can use this “extra” calculation to support a second light!  All that is required is to stick a second light-direction in the third set of texture coordinates, and to use a two-dimensional shadow texture that uses the same steps in the vertical direction as the horizontal direction.

Before we continue to refine our lighting scheme, we should insure that the quality of our per-pixel shading is optimal.  Our purpose in moving to pixel shader dot-products was to improve the look of large flat areas, and there is still a small problem with our method.  The look of these areas is vastly improved, but there are still times when the shadow creeps in a bit too far in the middle of polygons.  This is due to the linear interpolation of texture coordinates between vertexes.

While each normal is normalized to one unit at each vertex, the linear interpolation can cause an individual pixel to use a normal value that is less than one unit.  This can be seen visually in figure 2. Since the dot product of two vectors x and y is cos angle * (|x| + |y|)), a shorter vector means a lower result, and, therefore, an earlier sample in our texture.  This error can cause pixels that should fall on the light side of a “step” in our texture to fall on the dark side instead.

Fortunately, there is a solution to this problem, and it involves using a cube-map to generate our biased normals instead of grabbing them directly with texcoord. A cube-map is a combination of six textures that are placed on the six faces of a virtual cube. To sample a cube-map you pass in three texture coordinates in the -1 to 1 range.  These represent an x, y, and z direction from the center of the cube.  Sampling the cube-map projects a ray in this direction and results in the color of the pixel where the ray intersects the cube.

Using the knowledge of how cube-maps work, we can generate a special cube map that will take our interpolated normal (which fortunately points in the same direction as the correct normal) and output a biased RGB color representing the normalized normal.  We can use this result with our texm3x2 instructions as in the previous algorithm.  This solves our problem with short normals and improves the look of our stepped shading.  Listing 3 shows the modified code.

Listing 3

vs.1.1 ; modifications to previous vertex shader

; stick first light position in second stage

mov oT1, c[LIGHT_POSITION_A]

; stick second light position in third stage

mov oT2, c[LIGHT_POSITION_B]

ps.1.1 ; modifications to previous pixel shader

tex t0



; grab the normalized normal

; normalizing cube map is in t0

texm3x2pad t1, t0_bx2    ; dot with light a for u value

texm3x2tex t2, t0_bx2
; dot with light b for v value

; 2d shade texture is in stage t2
If you look closely at the shadow borders in the third incarnation of our pixel shader, you will notice that the border seems a bit fuzzy.  Some shadow pixels are visible in the highlight area, and some highlight pixels are visible in the shadow area.  The reasons for this are two fold.  First, there are inaccuracies in the vertex normal that are introduced by having to bias it into the 0 to 1 range and back again.  Second, the floating point precision in the pixel shader isn’t high enough to give us perfect results.  Fixing these problems requires more advanced hardware, and this fixing may not even be required.  Some people prefer the softer borders produced by these inaccuracies.

So far we have been concerned primarily with shading the model accurately, and we haven’t discussed light color.  It is important to realize that a cel-shader cannot support different colored lights illuminating the same model.  Multiple colored lights create a gradient effect that disrupts the stepped-shading of the object.  It is therefore necessary to combine the colors of the various lights into one value and modulate the resulting light color with the shading and the model color.  We can do this easily by putting the combined light value into a pixel shader constant.

A dark light-color will make the object appear dark, and a bright color will make the object appear bright.  It is possible to modulate x2 to allow bright lights to actually brighten the object above its diffuse color.  I have found that 1.5x overbrightening (modulate x2 with a 0.75 scaling factor) gives a very nice cartoon effect.

Unfortunately, the brightness created by light-color alone tends to look a bit static.  If the model has a pulsating light next to it, the whole model tends to brighten and darken with the light, while the shading stays exactly the same.  In a traditional cartoon world, the highlighted areas tend to expand as lights get brighter.

We can achieve this effect by scaling the light-direction vector with the brightness of the light.  The brightness of a light is equal to the brightness of its brightest color component.  Because result of the dot-product of two vectors is influenced by their magnitude, scaling the light-direction by brightness has the effect of biasing the shading towards highlights for brighter lights and shadows for dimmer lights.  With some experimenting with the texture, it is even possible to use two shades for dim lights (i.e. 0.5 or less), with a third shade that only appears near brighter lights.

Scaling the light-direction vector as in the above method has the added benefit of creating some distinction between our two lights.  While it is necessary to combine the colors of the two lights that illuminate the model, each light-direction can be scaled by its individual brightness, which allows one light to appear brighter than the other because its highlight areas are bigger while maintaining the stepped-shading look.

Inking

Now that we have our painter working, we’ll look at the inking portion of our cartoon renderer.  There are three different categories of lines that we want our inker to take care of.  The first type is an outline that surrounds each model, distinguishing it from surrounding objects that may be the same color.  The second type of line is drawn on sharp edges or creases in the model.  These lines emphasize the features of the model.  The third type of line is the characteristic line created by an artist.  These lines could be used on a model of a painting for example, or to emphasize a character’s facial features that aren’t distinct enough to generate an edge line.

A great deal of investigation has been done into world-space lining techniques.  Buchanan [3] recommended using an edge-buffer to keep track of edge polygons.  Raskar [4] developed methods of extending back-facing polygons to create outlines and edges. The researchers at nvidia [5] show how to use a vertex shader to generate outlines on a model.  Mitchell [6], was the first to suggest using modern hardware to create outlines by comparing pixels in image-space.

Despite the prevalence of world space techniques, edge-lines and and artist-lines almost always require the generation of new geometry and/or the maintenance of software information that doesn’t translate well to programmable hardware. The vertex-shader outlining technique is reasonable effective for high-poly, smooth models, but its quality is simply unacceptable for low-poly and angular models.  Programmable pixel shaders make image-space edge detection techniques possible at real-time frame rates.  We are therefore going to use image-space techniques for our cel-shader.

So, how do we use pixel shaders for image filtering?  First, it is necessary to put the source image in all four texture stages.  Then we create a vertex shader that offsets the texture coordinates slightly.  For example, we move the coordinates for the first stage up and to the left the equivalent of one pixel.  We move the third stage down and to the right and similarly offset the second and forth stage down-left and up-right, respectively. Listing 4 shows the code for a simple vertex program that does this.

Listing 4

// c++ code

// create 1 pixel offsets

float const kPerTexelWidth  = 1.0f/static_cast<float>(m_d3dsdBackBuffer.Width);

float const kPerTexelHeight = 1.0f/static_cast<float>(m_d3dsdBackBuffer.Height);

float s = 0.5f;

float const OffsetX[4] = { -s * kPerTexelWidth, 

                           -s * kPerTexelWidth,  

                            s * kPerTexelWidth,   

                            s * kPerTexelWidth  };

float const OffsetY[4] = { -s * kPerTexelHeight, 

                            s * kPerTexelHeight, 

                            s * kPerTexelHeight, 

                           -s * kPerTexelHeight };

// stick offsets in vertex shader constants

for (int i = 0; i < 4; ++i)

{

D3DXVECTOR4 Off(OffsetX[i], OffsetY[i], 0.0f, 0.0f);

m_pd3dDevice->SetVertexShaderConstantF(

                   TEXTURE_OFFSET + i, (float *)&Off, 1);

}

// specify the offset set to use

float fOffsetSet = TEXTURE_OFFSET;

m_pd3dDevice->SetVertexShaderConstantF(

                   OFFSET_SET, (float *)&fOffsetSet, 1);

; vertex shader code

vs.1.1

dcl_texcoord0 v1

mov a0.x, c[OFFSET_SET].x ; grab the proper offset

add oT0, v1, c[a0.x + 0]

add oT1, v1, c[a0.x + 1]

add oT2, v1, c[a0.x + 2]

add oT3, v1, c[a0.x + 3]

Now that we can sample four different pixels surrounding our output pixel, we can manipulate these pixels to create various image-filters.  It is possible to do sharpening, blurring, and even luminance edge detection in short pixel shader programs.  A pixel program runs once for each pixel in the render target and has only one output: the color of the pixel.  Multiple pixels in the original image can be sampled through the four sets texture coordinates and different colors outputted based on traditional image-filtering algorithms.

The only disadvantage of the image-filter approach is that it is now necessary to render every object in the scene twice: once, into the back buffer to create the cartoon shading and, again, into a texture so that lines can be generated. More modern hardware can use multiple render targets to do this in one pass, but our selected hardware doesn’t allow it.  For now, we have to be content with the superior quality of this technique, and take a 50% frame rate hit that will translate into lower poly in-game models than could otherwise be used.

The goal of our outlining technique is to generate a set of lines in a texture that is the same size as the back-buffer.  This texture will be applied to a single quad and rendered over the top of our shaded scene.  This doesn’t really constitute a third rendering pass, because only two triangles are sent to the hardware.  It can have an impact on frame rate as it pushes the hardware’s fill-rate, especially in high-resolution modes.  Remember that our pixel shader will have to run once for each pixel on the screen, so higher resolution means lower frame rate. Fortunately, the impact is fairly constant for each resolution, regardless of how many polygons are in the underlying scene.

First let’s concern ourselves with the true outlines, the lines that need to surround each object in the scene.  If we had some way of getting depth information for each pixel, we could compare neighboring depth values to see if a jump had occurred.  Unfortunately, the depth-buffer isn’t available from a pixel program, so it is necessary to encode the information into the texture we are rendering into.  For this we need a vertex shader program that will calculate the distance from the camera to each vertex and put it in the alpha of the vertexes diffuse color.  An alpha of zero will represent the near plane, and an alpha of One will represent the far plane.  It is useful to note that we could put the depth value into one of the color channels, but we will end up using the RGB channels for the information that is necessary to generate the other types of lines, so it is most efficient here.

Now that we have depth values in the alpha channels of our rendered texture, we can compare neighboring pixel values, and set the pixel black if the depth values are different or transparent if they are similar.  In our pixel shader we will compare upper-left pixel against the lower-right pixel and lower-left pixel against the upper-right pixel.  If there is a difference in either set, we are on the edge of an object and we need to output a line pixel.  Otherwise, we output a transparent pixel so that the shading shows through.

Next, we need to generate lines on the sharp edges of objects.  There is one characteristic that distinguishes edge vertexes from other vertexes.  The normals of two edge vertexes point in different directions.  If the normals point away from each other, we have found an edge.  If the normals point toward each other then we have found a crease.  In either case we want a black line to appear at the location.

In order to make the vertex normal information available to the pixel shader, it is necessary to encode the normal in such a way that neighboring pixels can be compared.  You may recall that the dot product is quite useful for discerning whether vectors are pointing in the same direction.  If we can bias the x,y,z values of the vertex normal and stick them into the vertex color RGB values then we can use the bx2 modifier to return them to the proper range and compare the values with the pixel shader dp3 instruction.

We did something quite similar to this when we needed vertex normals for our stepped shading, and we discovered that errors occurred when the normals were interpolated between vertexes.  We will suffer from the same errors here, so we should reuse our normalizing cube-map.  We can encode the normal as texture coordinates and look up the normalized, biased vector in the cube-map.  The RGB result needs to be combined with the depth alpha values that are in the diffuse color using the fixed-function pipeline or a simple pixel shader.

Our render texture now has the required info for depth and edge lines, now all we need is a pixel shader that we can enable for our overlay-quad that will turn this texture into lines for us.  Listing 5 shows a pixel shader that generates lines from this texture.  It depends on the one conditional instruction available in PS 1.1: cnd. This instruction will choose between two source colors and put them into r0, based on comparing r0.a with 0.5.  We can set a depth threshold and an edge threshold that will change how different the RGB and alpha values need to be in order to generate a line.

Listing 5

ps.1.1

def c0, 0.0f, 0.0f, 0.0f, 0.9f     ; threshold for normals 

def c1, 0.0f, 0.0f, 0.0f, 0.25f    ; depth threshold

def c2, 1.0f, 1.0f, 1.0f, 0.0f     ; white and transparent

#define C_LINE_COLOR     c5

tex t0

tex t1

tex t2

tex t3

dp3 r0.rgb, t0_bx2, t2_bx2         ; dot the normals for opposite pixels

+sub r0.a, t0.a, t2.a              ; find depth differences

dp3 r1.rgb, t1_bx2, t3_bx2         ; repeat for other set

+sub r1.a, t1.a, t3.a

sub_x4_sat r0.rgb, r0, c0.a        ; subtract the normal threshold and clamp

+mad r0.a, r0.a, r0.a, c1.a        ; square the differences and add threshold

sub_x4_sat r1.rgb, r1, c0.a        ; repeat for other set

+mad r1.a, r1.a, r1.a, c1.a

mul_x4_sat r0.rgb, r0, r1          ; combine the clamped normal values

+add r0.a, r0.a, r1.a              ; combine the differences

add r0.a, 1 - r0.b, r0.a           ; combine depth and reversed normal values

cnd r0, r0.a, C_LINE_COLOR, c2     ; set pixel line color or transparent

Of all the shaders we have looked at so far, this one went through the most iterations to get to this state.  Initially my cel-shader used two separate pixel shaders for edge and depth lines and it was still difficult to squeeze them into eight instructions. The added passes of multiple shaders was unacceptable, so I set about combining the shaders into one.  I was stuck on a ten instruction version for awhile, but with some modifications and creative co-issuing I managed to squeeze it down to seven instructions.

Staying one instruction below the limit is important, because we are going to use our eighth instruction to improve the look of our edge lines.  With the current version of the shader, some smooth areas of a model won’t generate a line close up, but a line will start appearing as the model gets farther away.  This is because two normals that point in different directions on a smooth surface will be many pixels apart when the object is close, but as the object moves farther from the eye, less pixels are used for the object.  Two neighboring pixels may suddenly have normals that are far enough apart to generate an edge line.

We can alleviate this problem somewhat by scaling the edge threshold with the depth value of one of the pixels.  If we decrease the edge threshold for distant objects, the smooth areas won’t suddenly become edges in the distance.  The changes to the pixel shader are shown in Listing 6.  You may need to play with the edge threshold and depth adjustment values.  Optimal values depend on resolution, far plane distance, and the general look of different models in your game.

Listing 6

ps.1.1 ; modifications to previous pixel shader

def c3, 0.0f, 0.0f, 0.0f, -0.25f    ; depth adjustment

mad t0.a, t0.a, c3.a, c0.a         ; scale the normal threshold with depth (uses first texture stage pixel)

sub_x4_sat r0.rgb, r0, t0.a        ; subtract the normal threshold and clamp

+mad r0.a, r0.a, r0.a, c1.a        ; square the differences and add threshold

sub_x4_sat r1.rgb, r1, t0.a        ; repeat for other set

+mad r1.a, r1.a, r1.a, c1.a

If we had an infinite polygon limit, we might be happy with the current state of the cartoon renderer.  Since lines are generated automatically from edges, we can create lines by adding polygons to the model.  Unfortunately, this quickly becomes impractical for games.  To keep our frame rates high, we have to be able to use lower poly models and be able to place detail lines by hand on the model.  These are the artist-specified lines we discussed earlier.  The first method that comes to mind is to simply draw lines on the texture that is applied to the model.  This works, but the quality is unacceptable for most games, because the lines change in width as one gets close to the model.  This creates a sharp contrast with the other lines in the game and breaks the consistency of the cartoon look and feel.

Ideally what we need is a way to hand-generate lines using a similar method to the ones we used for edge and depth lines.  It would be even better if we could use the same pixel shader that we are already using for these other lines so as to minimize rendering passes.  This is, in fact, possible, due to the fact that only about half of the colors representing vertex normals can be in view at any one time.  We are going to develop two different ways for an artist to specify lines that will work with our current pixel shader.

Assuming backface culling is enabled, a visible vertex normal’s end-point will be on the positive side of a plane that is perpendicular to and facing the eye position that passes through the vertex.  This means that all of the colors that represent normals that point to the negative side of the plane can be used by our artist-line algorithm without interfering with the normal lines.  First, we need to generate a set of normals that are equidistant from a plane (representing the limit of the set of visible normals) and each other. We then need to rotate these normals away from the eye position and encode them as colors.

Our first method for artist-lines uses vertex-color information in the model itself.  We pick four colors that represent equidistant normals from the x-y plane and paint polygons in the model with these colors. An example of four possible vectors and colors is shown in figure 3.  These four colors should be enough to generate lines on even complex models if they are used wisely.  Lines will appear in between polygons that we have painted with different colors.  Slightly more accurate results could be obtained by using the four normals directly without encoding them as a color, but this requires writing special tools to make it easy for artists to put these lines on the model and to export the extra normal data for each vertex.

In our renderer, a vertex shader converts the colors into normals and rotated so they point in the same direction as the eye-to-vertex vector.  This generates normals that are always pointing away from the eye and won’t interfere with our edge lines.  Listing 7 shows a vertex shader that performs this rotation. Our current pixel shader will automatically generate lines from this data.

Listing 7

vs.1.1

dcl_position v0

dcl_color0 v1

dcl_normal v3

def c30, 0.0f, 0.9f, 0.1f, 2000.0f ; depth modifiers--y and z used to clamp values to 0.1-1.0, w is far plane

def c31, 0.0f, 0.0f, 2.0f, -1.0f ; bias

def c32, 0.0f, 1.0f, 0.0f, 1.0f ; up vector

; grab eye to vertex vector

add r2, v0, -c[EYE_POSITION]

dp3 r2.w, r2, r2

rsq r2.w, r2.w

; stick offset distance from camera in oD0 alpha

mul r3.w, r2.w, c30.w

rcp r3.w, r3.w

mad oD0, r3.w, c30.x, c30.y

; calculate new coordinate system into r4-r6

mul r6, r2, r2.w

; z-axis

; x-axis = up x z-axis( 3-vector cross-product)

mul r4, c32.yzxw, r6.zxyw

mad r4, -r6.yzxw, c32.zxyw, r4

; normalize x-axis

dp3 r4.w, r4, r4

rsq r4.w, r4.w

mul r4, r4, r4.w

; y-axis = z-axis x x-axis

mul r5, r6.yzxw, r4.zxyw

mad r5, -r4.yzxw, r6.zxyw, r5

; change vertex color to normal

mad r7, v1, c31.z, c31.w

; transpose and multiply

; transpose is necessary because we are going from new coordinate system into old

mul r0, r4, r7.x

mad r0, r5, r7.y, r0

mad r0, r6, r7.z, r0

mov oT0, r0

Generating lines from vertex colors is a fantastic achievement, but it would be even better if we could specify lines in a texture.  This will allow us to generate lines across polygons, and not just in between them. We are going to use a similar method to make this possible.  Unfortunately, texture information isn’t available in the vertex shader, so we’ll have to be creative in the pixel shader.  To make our four vectors available, we are going to generate a dynamic 2x2 texture that contains the encoded version of our four colors.  The normals will be rotated to face the same direction as an eye-to-object vector, encoded, and placed in the texture in every frame.  This isn’t going to be quite as accurate as the per-vertex method, because we aren’t rotating the vectors per-vertex.  When we are close to large objects, some lines may disappear between texture-generated-line sections of the object and normal-generated-line sections.

For an artist to create texture lines, she makes a texture that uses the following four RGBA colors:

(1.0, 0.0, 0.0, 0.0)

(1.0, 0.0, 0.0, 1.0)

(0.0, 0.0, 0.0, 0.0)

(0.0, 0.0, 0.0, 1.0)

It is then possible to sample our dynamic 2x2 texture using the texreg2ar instruction.  This instruction interprets the red and alpha values of a supplied texture as texture coordinates to sample a second texture.  This essentially replaces the four different colors from the source texture with our 4 normal values.  Lines will automatically be generated by our previous pixel shader.  These lines will always remain the same width, but may become a bit jagged when viewed extremely closely.  The jaggedness is based on the texture resolution, and higher-resolution texture will generally solve quality problems.

It bears noting that using our optimized methods for generating lines, any given set of polygons can only use one method that uses the RGB color.  Each polygon can use either edge lines, vertex color hand lines, or texture hand lines.  Depth lines will work with any of the methods.  I have found that the best way to specify which style to use is to include it in the material.  That way, a given object can use multiple styles for different sets of its polygons.  A character could use edge lines for its body for example, and texture lines for the face where details could be placed by hand. 
It is common for games to use partially-transparent textures for objects like fences and leaves, where modeling the objects in geometry would increase the polygon count dramatically.  This will work with our outlining techniques as long as we render the object with alpha into our overlay texture.  The best method for this is to use 1-bit alpha in a texture and turn on alpha-testing.  Zero alpha pixels will be discarded and lines will be drawn around areas that remain visible.

One very useful addition to our inker would be the ability to change line width.  The easiest way to do this is to change the size of our render texture.  A texture that is twice the size of the back buffer creates nice thin lines, assuming your hardware can support render textures that size.  Smaller textures can create thicker lines, but these lines tend to look more pixilated as they thicken.

Another approach to thicken lines is to render the overlay lines into a second render texture instead of the back-buffer.  We can then use another image-filtering pixel shader when we render this new texture to thicken the lines.  This method is quite fast for low- resolution output, but can be limited by fill-rate in high-resolution modes. Listing 8 contains two pixel shaders that use slightly different algorithms to thicken our original lines.  The first pixel shader thickens the line by outputting a line-pixel if any of the four sampled pixels are line pixels.  The second pixel shader creates smoothed thicker lines by linearly interpolating between the line-color and transparent depending on how many neighboring pixels are set.

Listing 8

ps.1.1 ; dilation pixel shader

def c0, 0.0f, 0.0f, 0.0f, 0.25f    ; for 1/4 multiplication

def c1, 0.0f, 0.0f, 0.0f, -0.75f   ; for subtracting 

def c2, 1.0f, 1.0f, 1.0f, 0.0f     ; white and transparent

#define C_LINE_COLOR     c5

tex t0

tex t1

tex t2

tex t3

mul r0.a, 1 - t0.a, c0.a           ; sum 1/4 of inverse samples

mad r0.a, 1 - t1.a, c0.a, r0.a     ; alpha of lines are 1.0 alpha

mad r0.a, 1 - t2.a, c0.a, r0.a

mad r0.a, 1 - t3.a, c0.a, r0.a

add_x4 r0.a, r0.a, c1.a            ; subtract .75 and mul * 4 (only transparent pixels will remain transparent)

cnd r0, r0.a, c2, C_LINE_COLOR     ; conditionally choose between transparent and line color

ps.1.1 ; smooth dilation pixel shader

def c0, 0.0f, 0.0f, 0.0f, 0.5f     ; smoothing threshhold 0.25 = max smoothing 1.0 = no smoothing

def c2, 1.0f, 1.0f, 1.0f, 0.0f     ; white and transparent

#define C_LINE_COLOR
c5

tex t0

tex t1

tex t2

tex t3

mul r0.a, t0.a, c0.a               ; combine the four samples using threshold and clamp

mad r0.a, t1.a, c0.a, r0.a

mad r0.a, t2.a, c0.a, r0.a

mad_sat r0.a, t3.a, c0.a, r0.a

mov r1.rgb, C_LINE_COLOR           ; create a zero alpha version of line color

+ mov r1.a, c2

lrp r0, r0.a, C_LINE_COLOR, r1     ; interpolate between alpha and no alpha

